

NAG Fortran Library Routine Document

G07CAF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

G07CAF computes a t -test statistic to test for a difference in means between two Normal populations, together with a confidence interval for the difference between the means.

2 Specification

```

SUBROUTINE G07CAF(TAIL, EQUAL, NX, NY, XMEAN, YMEAN, XSTD, YSTD, CLEVEL,
1              T, DF, PROB, DL, DU, IFAIL)
INTEGER        NX, NY, IFAIL
real          XMEAN, YMEAN, XSTD, YSTD, CLEVEL, T, DF, PROB, DL, DU
CHARACTER*1    TAIL, EQUAL

```

3 Description

Consider two independent samples, denoted by X and Y , of size n_x and n_y drawn from two Normal populations with means μ_x and μ_y , and variances σ_x^2 and σ_y^2 respectively. Denote the sample means by \bar{x} and \bar{y} and the sample variances by s_x^2 and s_y^2 respectively.

G07CAF calculates a test statistic and its significance level to test the null hypothesis $H_0 : \mu_x = \mu_y$, together with upper and lower confidence limits for $\mu_x - \mu_y$. The test used depends on whether or not the two population variances are assumed to be equal.

1. It is assumed that the two variances are equal, that is $\sigma_x^2 = \sigma_y^2$.

The test used is the two sample t -test. The test statistic t is defined by;

$$t_{obs} = \frac{\bar{x} - \bar{y}}{s \sqrt{(1/n_x) + (1/n_y)}}$$

where $s^2 = ((n_x - 1)s_x^2 + (n_y - 1)s_y^2)/(n_x + n_y - 2)$ is the pooled variance of the two samples.

Under the null hypothesis H_0 this test statistic has a t -distribution with $(n_x + n_y - 2)$ degrees of freedom.

The test of H_0 is carried out against one of three possible alternatives;

$H_1 : \mu_x \neq \mu_y$; the significance level, $p = P(t \geq |t_{obs}|)$, i.e., a two-tailed probability.

$H_1 : \mu_x > \mu_y$; the significance level, $p = P(t \geq t_{obs})$, i.e., an upper tail probability.

$H_1 : \mu_x < \mu_y$; the significance level, $p = P(t \leq t_{obs})$, i.e., a lower tail probability.

Upper and lower $100(1 - \alpha)\%$ confidence limits for $\mu_x - \mu_y$ are calculated as:

$$(\bar{x} - \bar{y}) \pm t_{1-\alpha/2} s \sqrt{(1/n_x) + (1/n_y)}.$$

where $t_{1-\alpha/2}$ is the $100(1 - \alpha/2)$ percentage point of the t -distribution with $(n_x + n_y - 2)$ degrees of freedom.

2. It is not assumed that the two variances are equal.

If the population variances are not equal the usual two sample t -statistic no longer has a t -distribution and an approximate test is used.

This problem is often referred to as the Behrens–Fisher problem, see Kendall and Stuart (1969). The test used here is based on Satterthwaites procedure. To test the null hypothesis the test statistic t' is used where

$$t'_{obs} = \frac{\bar{x} - \bar{y}}{se(\bar{x} - \bar{y})}$$

$$\text{where } se(\bar{x} - \bar{y}) = \sqrt{\frac{s_x^2}{n_x} + \frac{s_y^2}{n_y}}.$$

A t -distribution with f degrees of freedom is used to approximate the distribution of t' where

$$f = \frac{se(\bar{x} - \bar{y})^4}{\frac{s_x^2/n_x^2}{(n_x - 1)} + \frac{s_y^2/n_y^2}{(n_y - 1)}}.$$

The test of H_0 is carried out against one of the three alternative hypotheses described above, replacing t by t' and t_{obs} by t'_{obs} .

Upper and lower $100(1 - \alpha)\%$ confidence limits for $\mu_x - \mu_y$ are calculated as:

$$(\bar{x} - \bar{y}) \pm t_{1-\alpha/2} se(x - \bar{y}).$$

where $t_{1-\alpha/2}$ is the $100(1 - \alpha/2)$ percentage point of the t -distribution with f degrees of freedom.

4 References

Johnson M G and Kotz A (1969) *The Encyclopedia of Statistics* 2 Griffin

Kendall M G and Stuart A (1969) *The Advanced Theory of Statistics (Volume 1)* (3rd Edition) Griffin

Snedecor G W and Cochran W G (1967) *Statistical Methods* Iowa State University Press

5 Parameters

- 1: TAIL – CHARACTER*1 Input
On entry: indicates which tail probability is to be calculated, and thus which alternative hypothesis is to be used.
 If TAIL = 'T', the two tail probability, i.e., $H_1 : \mu_x \neq \mu_y$.
 If TAIL = 'U', the upper tail probability, i.e., $H_1 : \mu_x > \mu_y$.
 If TAIL = 'L', the lower tail probability, i.e., $H_1 : \mu_x < \mu_y$.
Constraint: TAIL = 'T', 'U' or 'L'.
- 2: EQUAL – CHARACTER*1 Input
On entry: indicates whether the population variances are assumed to be equal or not.
 If EQUAL = 'E', the population variances are assumed to be equal, that is $\sigma_x^2 = \sigma_y^2$.
 If EQUAL = 'U', the population variances are not assumed to be equal.
Constraint: EQUAL = 'E' or 'U'.
- 3: NX – INTEGER Input
On entry: the size of the X sample, n_x .
Constraint: NX \geq 2.

- 4: NY – INTEGER Input
On entry: the size of the Y sample, n_y .
Constraint: $NY \geq 2$.
- 5: XMEAN – *real* Input
On entry: the mean of the X sample, \bar{x} .
- 6: YMEAN – *real* Input
On entry: the mean of the Y sample, \bar{y} .
- 7: XSTD – *real* Input
On entry: the standard deviation of the X sample, s_x .
Constraint: $XSTD > 0.0$.
- 8: YSTD – *real* Input
On entry: the standard deviation of the Y sample, s_y .
Constraint: $YSTD > 0.0$.
- 9: CLEVEL – *real* Input
On entry: the confidence level, $1 - \alpha$, for the specified tail. For example CLEVEL = 0.95 will give a 95% confidence interval.
Constraint: $0.0 < \text{CLEVEL} < 1.0$.
- 10: T – *real* Output
On exit: contains the test statistic, t_{obs} or t'_{obs} .
- 11: DF – *real* Output
On exit: contains the degrees of freedom for the test statistic.
- 12: PROB – *real* Output
On exit: contains the significance level, that is the tail probability, p , as defined by TAIL.
- 13: DL – *real* Output
On exit: contains the lower confidence limit for $\mu_x - \mu_y$.
- 14: DU – *real* Output
On exit: contains the upper confidence limit for $\mu_x - \mu_y$.
- 15: IFAIL – INTEGER Input/Output
On entry: IFAIL must be set to 0, -1 or 1. Users who are unfamiliar with this parameter should refer to Chapter P01 for details.
On exit: IFAIL = 0 unless the routine detects an error (see Section 6).
For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, for users not familiar with this parameter the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

6 Error Indicators and Warnings

If on entry $IFAIL = 0$ or -1 , explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

$IFAIL = 1$

On entry, $TAIL \neq 'T', 'U'$ or $'L'$,
 or $EQUAL \neq 'E'$ or $'U'$,
 or $NX < 2$,
 or $NY < 2$,
 or $XSTD \leq 0.0$,
 or $YSTD \leq 0.0$,
 or $CLEVEL \leq 0.0$,
 or $CLEVEL \geq 1.0$.

7 Accuracy

The computed probability and the confidence limits should be accurate to approximately 5 significant figures.

8 Further Comments

The sample means and standard deviations can be computed using G01AAF.

9 Example

The following example program reads the two sample sizes and the sample means and standard deviations for two independent samples. The data is taken from page 116, Snedecor and Cochran (1967) from a test to compare two methods of estimating the concentration of a chemical in a vat. A test of the equality of the means is carried out first assuming that the two population variances are equal and then making no assumption about the equality of the population variances.

9.1 Program Text

Note: the listing of the example program presented below uses *bold italicised* terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```
*      G07CAF Example Program Text
*      Mark 15 Release. NAG Copyright 1991.
*      .. Parameters ..
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
*      .. Local Scalars ..
      real             CLEVEL, DF, DL, DU, PROB, T, XMEAN, XSTD, YMEAN,
+                    YSTD
      INTEGER          IFAIL, NX, NY
*      .. External Subroutines ..
      EXTERNAL         G07CAF
*      .. Executable Statements ..
      WRITE (NOUT,*) 'G07CAF Example Program Results'
*      Skip heading in data file
      READ (NIN,*)
      READ (NIN,*) NX, NY
      READ (NIN,*) XMEAN, YMEAN, XSTD, YSTD
      READ (NIN,*) CLEVEL
      IFAIL = 0

*      CALL G07CAF('Two','Equal',NX,NY,XMEAN,YMEAN,XSTD,YSTD,CLEVEL,T,DF,
+                PROB,DL,DU,IFAIL)
*
```

```

WRITE (NOUT,*)
WRITE (NOUT,*) 'Assuming population variances are equal.'
WRITE (NOUT,*)
WRITE (NOUT,99999) 't test statistic = ', T
WRITE (NOUT,99998) 'Degrees of freedom = ', DF
WRITE (NOUT,99997) 'Significance level = ', PROB
WRITE (NOUT,99999)
+ 'Lower confidence limit for difference in means = ', DL
WRITE (NOUT,99999)
+ 'Upper confidence limit for difference in means = ', DU
WRITE (NOUT,*)
IFAIL = 0
*
CALL G07CAF('Two','Unequal',NX,NY,XMEAN,YMEAN,XSTD,YSTD,CLEVEL,T,
+          DF,PROB,DL,DU,IFAIL)
*
WRITE (NOUT,*)
WRITE (NOUT,*) 'No assumptions about population variances .'
WRITE (NOUT,*)
WRITE (NOUT,99999) 't test statistic = ', T
WRITE (NOUT,99997) 'Degrees of freedom = ', DF
WRITE (NOUT,99997) 'Significance level = ', PROB
WRITE (NOUT,99999)
+ 'Lower confidence limit for difference in means = ', DL
WRITE (NOUT,99999)
+ 'Upper confidence limit for difference in means = ', DU
STOP
*
99999 FORMAT (1X,A,F10.4)
99998 FORMAT (1X,A,F8.1)
99997 FORMAT (1X,A,F8.4)
END

```

9.2 Program Data

G07CAF Example Program Data

```

4 8
25.0 21.0
0.8185 4.2083
0.95

```

9.3 Program Results

G07CAF Example Program Results

Assuming population variances are equal.

```

t test statistic =      1.8403
Degrees of freedom =      10.0
Significance level =      0.0955
Lower confidence limit for difference in means =      -0.8429
Upper confidence limit for difference in means =      8.8429

```

No assumptions about population variances .

```

t test statistic =      2.5922
Degrees of freedom =      7.9925
Significance level =      0.0320
Lower confidence limit for difference in means =      0.4410
Upper confidence limit for difference in means =      7.5590

```
